

# Stylish Waterfall Graphs using SAS® 9.3 and 9.4 Graph Template Language

Setsuko Chiba, Exelixis Inc. South San Francisco, CA

## ABSTRACT

One stylish graph provides a clear picture of data summaries reported on tables and listings. Waterfall graphs show at a glance the increase/decrease of data analysis results from various industries. The introduction of SAS® 9.2 Statistical Graphics (SG) procedures enabled SAS programmers of all skill levels to produce high quality results with simple coding. Moreover, SAS programmers can create sophisticated graphs in stylish custom layouts using SAS® 9.3 and later Graph Template Language (GTL) with Output Delivery System (ODS) Style Template. This poster presents two sets of example waterfall graphs in clinical trials using SAS 9.3 and later. Example 1 for presentation: displays colorful graphs using SAS® 9.3 and 9.4 new options. Example 2 for publication: displays simple graphs with Gray Scale Color Coding and patterns. You can create these graphs on Unix or Windows.


## INTRODUCTION

SAS(R) 9.4 ODS Graphics: Procedures Guide, Second Edition describes the waterfall chart as follows: "In the chart, bars represent an initial value of Y and a series of intermediate values identified by X leading to a final value of Y." In Oncology Clinical Trials, the waterfall graphs are usually used for final analysis comparing drug A vs. placebo. As Figure 1 shows, Y-axis: a vertical bar represents the final change or % change from baseline in the tumor size per subject on the X-axis. Values on the left side are worst and on the right side are best in terms of percent change observed in two groups of subjects. The entire plot looks like the waterfall running high to low.

### Example 1 Waterfall Plot for Presentation: Figure 1

SAS Global Inc.  
Protocol Washington

Development



Note: y= -70 indicates the maximum tumor size reduction.



## WATERFALL PLOTS BY GTL

From my personal experience using SAS 9.2 and later for graphs, I almost always started using the existing sample SAS code prepared for SGPLOT or SGPANEL, but I found I always completed assignments with the help of GTL. In order to meet the specific request that the graphs appeal to the audience, I had to customize graphical elements such as style, color, text, font, location etc., beyond SG procedures. The waterfall plots in this poster were originally produced by GTL 9.3 and modified based on new features of SAS 9.4. Please refer to "What's New In SAS 9.4 Graph Template Language" for details.

Figure 1 was generated first creating **PROC TEMPLATE** called **Waterfall**. Next, **PROC RENDER** was used to create the above output. Simple **Sample data** with 3 variables were generated for this poster. The y variable was ordered by high to low (descending order) and each subject was assigned x values from 1 to n based on y value of subject in each group (Drug\_A or Placebo).

Special attention can be paid using **Layout Gridded** to generate two rectangular cells stacked together in the graph: **upper cell** and **lower cell**. Similarly, **Layout Gridded** was used to generate a small box showing statistics inside each cell.

1. **Entry title** statement with option **textattrs** = (color=green) was used to define the title partially in color and left aligned using **haling** = left (SAS 9.4).
2. **Layout Gridded** with option **columns=1** was used to create two rectangular frames of the same size, with Drug\_A on the top and Placebo on the bottom of Figure 1. **Layout Gridded** statement needs to be outside of **Layout Overlay** since the two rectangular boxes are not nested inside each other.
3. **Layout Overlay** includes two other plots: **Barchart** and **Scatterplot** which share the common X and Y axis in each cell.
  - **Barchart** was preferred to **Waterfallchart** (SAS 9.3) as categorical plot. The appearance of long vertical lines was preferred rather than the steps created by SAS 9.3. The variables: **category** and **response** replaced original X and Y based on the recommendation by SAS 9.4 New Enhancement. The option **dataskin** was used to add shining effects on the graphs (SAS 9.4 fully supports Barchart).
4. **Scatterplot** with options **Legendlabel** and **Markerattrs** was used to create a red Star Symbol indicating the best performance.
5. **Layout Gridded** inside of **Layout Overlay** with 3 **Entry** statements created statistics of each treatment for **n=**, **mean=** and **max=** (the best size reduction) inside of the small box. Macro variables were used to pass the values: **&\_n1**, **&\_mean1** and **&\_min1** for Drug\_A as well as **&\_n2**, **&\_mean2** and **&\_min2** for Placebo.
6. The above steps: 3 - 5 were repeated for the lower cell between **Layout Overlay** and **Endlayout** statement.
7. **Discretelegend** was used describing the **symbol star** as "Best".
8. **Drawtext** statement was used to create watermark "SAS Global Forum 2014" as background with option **Transparency (.75)** and **Rotate (-35)** instead of **SGANNO** data set (SAS 9.4) because the coding was easier and produced the same result.



## PROC TEMPLATE for Figure 1

```

PROC TEMPLATE;
  Define statgraph waterfall;
  Begingraph/;
  Entrytitle halign=left   textattrs=(color=green) "Figure 1"
             halign=center textattrs=(color=black family="Albany AMT" )
  "Waterfall Plot: The Best Percentage Change in Tumor Size from Baseline ";;
  Entryfootnote halign=left "Note: y= &_min1 indicates the maximum tumor
                           size reduction.";
  Layout Gridded / columns=1;
      *-----;
      *- Upper Cell: Drug_A --;
      *-----;

  Layout Overlay/
    xaxisopts=(griddisplay=on display=(label) label="Drug_A ")
    yaxisopts=(griddisplay=on display=all label=" " offsetmin=0.2
              offsetmax=0.2);
  Barchart category= _xnum1 response= _ynum1/ fillattrs=(color=navy)
  dataskin=pressed;
  Referenceline y= -50 / lineattrs=(color=black);
  Scatterplot x= _xnum1 y= _ymark1 / name="mark1"
  Legendlabel="Best: &_min1" markerattrs=(symbol=starfilled color= red );
      Layout Gridded / rows=3 order=columnmajor border=true
      autoalign=(topright topleft);
      entry textattrs=(size=10) halign=right "N = &n1";
      entry textattrs=(size=10) halign=right "Mean = &_mean1 ";
      entry textattrs=(size=10) halign=right "Max. Change = &_min1";
  Endlayout;
  Discretelegend "mark1"/ location=inside halign=right valign=bottom;
  Endlayout;
      *-----;
      *- Lower Cell: Placebo --;
      *-----;

  Layout Overlay/ cycleattrs=true
    xaxisopts=( griddisplay=on display=(label) label="Placebo ")
    yaxisopts=( griddisplay=on display=all label=" "
              offsetmin=0.2 offsetmax=0.2 );
  Barchart category= _xnum2 response= _ynum2/
    fillattrs=(color= pink) dataskin=pressed;
  Referenceline y= -50 / lineattrs=(color=black);
      Layout Gridded / rows=3 order=columnmajor border=true
      autoalign=(topright topleft);
      entry textattrs=(size=10) halign=right "N = &n2";
      entry textattrs=(size=10) halign=right "Mean = &_mean2 ";
      entry textattrs=(size=10) halign=right "Max. Change = &_min2";
  Endlayout;
  Endlayout;
  Endlayout;
  Drawtext textattrs=(color=grey size=50pt) "SAS Global Forum 2014"/
    transparency=.75 rotate=-35 width=110 widthunit=percent justify=center;
  Endgraph;
End;

PROC SGRENDER data=water1 template=waterfall;
run;

```



## Example 2 Waterfall Plot for Publication: Figure 2\_a and Figure 2\_b





Figure 2\_a and Figure 2\_b were generated using **Gray\_Scale Color Codes** for publication. The figures represent the best 5 performance in 4 locations: USA, Europe, Asia and Africa. The **sample data** (Appendix A) was prepared by adding another variable called **location** to the original sample data used for Figure 1. The best 5 values were kept in each location. **PROC TEMPLATE Water** as well as **Style TEMPLATE Water2** were created based on **styles.journal3** as parent. **Barchart** with option **Display= (fill outline)** was used to create Figure 2\_a, and **Display= (fill fillpattern outline)** for Figure 2\_b. Complete SAS code will be provided as requested.



```

define style water2;
  parent=styles.journal3;
  *-----;
  *Figure 2_a showing Gray Colors;
  *-----;
  ** block when create 2_b;
  class GraphData1/ color=black;
  class GraphData2/ color=dimgray;
  class GraphData3/ color=lightgray;
  class GraphData4/ color=white;
  *-----;
  *Figure 2_b showing Patterns;
  *-----;
  ** block when create 2_a;
  *class GraphData1/ fillpattern="R3" color=dimgray;
  *class GraphData2/ fillpattern="X1" color=darkgray;
  *class GraphData3/ fillpattern="X5" color=lightgray;
  *class GraphData4/ fillpattern="L4" color=lightgray;
end;

```

The Figure 2\_a and figure 2\_b were created in **PDF** with option **STYLE=JOURNAL3**.

## CONCLUSION

As long as you can manage your job creating graphs using SG procedures like SGPLOT and SGPANEL, you will be fine. If you run into limitations using SG procedures more than twice, it is time for you to learn GTL. Once you developed a PROC TEMPLATE, you can produce impressive stylish graphs with different data by running PROC SGRENDER over and over. GTL will empower you to control the elements associated with the graphic display. Since the introduction of SAS 9.2 SG procedures and SAS 9.3 and later GTL, I have so much fun producing challenging graphs.

## REFERENCES, RECOMMENDED READING

SAS(R) 9.4 GRAPH TEMPLATE LANGUAGE: REFERENCE, SECOND EDITION, WHAT'S NEW IN SAS 9.4

[support.sas.com/documentation/cdl/en/grstatgraph/66421/HTML/default/viewer.htm#grstatgraphwhatsnew94.htm](http://support.sas.com/documentation/cdl/en/grstatgraph/66421/HTML/default/viewer.htm#grstatgraphwhatsnew94.htm)

GRAPH TEMPLATE LANGUAGE WHAT'S NEW IN SAS®9.4 2013, SAS INSTITUTE INC., CARY, NC, USA ISBN 978-1-61290-737-6 (ELECTRONIC BOOK)

Matange, Sanjay (2013-10-04). Getting Started with the Graph Template Language in SAS: Examples, Tips, and Techniques for Creating Custom Graphs . SAS Institute. Kindle Edition.

## CONTACT INFORMATION

Setsuko Chiba  
 Exelixis Inc.  
 210 East Grand Ave. South San Francisco  
 650-837-7176  
[schiba@exelixis.com](mailto:schiba@exelixis.com)

SAS and all other SAS Institute Inc. product or service names are registered trademarks or trademarks of SAS Institute Inc. in the USA and other countries. ® indicates USA registration.

Other brand and product names are trademarks of their respective companies.



# Appendix

Data Best5 for Figure 2\_a and Figure 2\_b

SAS Universal Viewer - [c:\users\schiba\desktop\best5.sas7bdat]

File Tools Window Help

Address

Library BEST5

Table View

Freeze Hide Show... Format Filter... Font... Find

	_grpn	_newsbj	Location	_ymark
1	1	233	1	-70
2	1	229	1	
3	1	224	1	
4	1	225	1	
5	1	220	1	
6	1	230	2	-44
7	1	226	2	
8	1	221	2	
9	1	218	2	
10	1	205	2	
11	1	227	3	-43
12	1	223	3	
13	1	219	3	
14	1	217	3	
15	1	209	3	
16	1	232	4	-51
17	1	231	4	
18	1	228	4	
19	1	222	4	
20	1	215	4	

Rows 1-20 of 20 Filter: Off Sort: none